

Syntax overview TutorialQL

<https://github.com/cwi-swat/rascal-ql-tutorial-prototype>

<http://www.rascal-mpl.org>

<http://docs.rascal-mpl.org/unstable/TutorHome/>

Context-free syntax

```
start syntax Form = form: "form" Id name "{" Question* questions "}";
```

```
syntax Question
= question: Label label Id name ":" QType tipe
| computed: Label label Id name ":" QType tipe "=" Expr expr
| ifThen: "if" "(" Expr cond ")" Question body () !>> "else"
| ifThenElse: "if" "(" Expr cond ")" Question body "else" Question elseBody
| @Foldable group: "{" Question* questions "}";
```

```
syntax Expr
= var: Id name
| integer: Integer
| string: String
| money: Money
| \true: "true"
| \false: "false"
| bracket "(" Expr ")"
> not: "!" Expr
> left (
    mul: Expr "*" Expr
  | div: Expr "/" Expr
)
> left (
    add: Expr "+" Expr
  | sub: Expr "-" Expr
)
> non-assoc (
    lt: Expr "<" Expr
  | leq: Expr "<=" Expr
  | gt: Expr ">" Expr
  | geq: Expr ">=" Expr
  | eq: Expr "==" Expr
  | neq: Expr "!=" Expr
)
> left and: Expr "&&" Expr
> left or: Expr "||" Expr;
```

```
lexical Label = @category="Constant" label: String;
```

```
syntax QType
= boolean: "boolean"
| string: "string"
| integer: "integer"
| money: "money";
```

Lexical syntax

```
lexical String = [""] StrChar* [""];
```

```
lexical StrChar = ![\"\\] | [\\][\\"nfbtr];
```

```
lexical Integer = [0-9]+ !>> [0-9];
```

```

lexical Money = [0-9]+ "." [0-9]+ !>> [0-9];

layout Standard = WhitespaceOrComment* !>> [\\t\\n\\f\\r] !>> "//";

lexical Comment = @category="Comment" "/" ![\n\r]* $;

syntax WhitespaceOrComment = whitespace: Whitespace | comment: Comment;

lexical Whitespace
  = [\\u0009-\\u000D \\u0020 \\u0085 \\u00A0 \\u1680 \\u180E \\u2000-\\u200A \\u2028 \\u2029 \\u202F \\u205F \\u3000];

lexical Id
  = id: ([a-z A-Z 0-9 _] !<< [a-z A-Z][a-z A-Z 0-9 _]* !>> [a-z A-Z 0-9 _]) \\ Keywords;

```

Abstract syntax

```

data Form
  = form(str name, list[Question] body);

data Question
  = question(Label label, Id name, QType tipe)
  | computed(Label label, Id name, QType tipe, Expr expr)
  | ifThen(Expr cond, Question body)
  | ifThenElse(Expr cond, Question body, Question elseBody)
  | group(list[Question] questions)
  ;

data Expr
  = var(Id id)
  | integer(int intValue)
  | string(str strValue)
  | money(real realValue)
  | \\true()
  | \\false()
  | not(Expr arg)
  | mul(Expr lhs, Expr rhs)
  | div(Expr lhs, Expr rhs)
  | add(Expr lhs, Expr rhs)
  | sub(Expr lhs, Expr rhs)
  | eq(Expr lhs, Expr rhs)
  | neq(Expr lhs, Expr rhs)
  | lt(Expr lhs, Expr rhs)
  | gt(Expr lhs, Expr rhs)
  | leq(Expr lhs, Expr rhs)
  | geq(Expr lhs, Expr rhs)
  | and(Expr lhs, Expr rhs)
  | or(Expr lhs, Expr rhs)
  ;

data QType
  = boolean()
  | integer()
  | money()
  | string()
  ;

data Id = id(str name);
data Label = label(str label);

```